Point symmetries of heat equation

Run NTMF064.Package.m first!

```
Clear["Global`*"]
```

Variables and differential equations in the form R(x,u,∂u,...) = 0

```
(* Independent variables *)
IndepVar = {x, t};
(* Dependent variables *)
DepVar = {u};
(* PDEs, only the functions R(...) without "== 0" *)
PDEs = {D[u[x, t], t] - D[u[x, t], x, x]}
```

$\{u^{(0,1)}[x, t] - u^{(2,0)}[x, t]\}$

Expression to substitute for in the infinitesimal criterion of invariance, when dealing with the heat equation all time derivatives can be replaced by space derivatives

```
subs = {D[u[x, t], t]};
sol = Solve[PDEs == 0, subs]
```

$\{\{u^{(0,1)}[x, t] \to u^{(2,0)}[x, t]\}\}$

Finding point symmetries by using a more and more specific ansatz

General ansatz

```
(* Infinitesimals for all variables *)
ξ[t] = Ξt[x, t, u[x, t]];
ξ[x] = Ξx[x, t, u[x, t]];
η[u] = H[x, t, u[x, t]];
(* Next expression should return zeroes
 if infinitesimals give a point symmetry of PDEs *)
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

```
u[0] = u[x, t]
u[1] = u^(0,1)[x, t]
u[2] = u^(1,0)[x, t]
u[3] = u^(0,2)[x, t]
u[4] = u^(1,1)[x, t]
u[5] = u^(2,0)[x, t]
```

$$\{2\,u[2]\,u[4]\,\Xi t^{(0,0,1)}[x,t,u[0]] + 2\,u[2]\,u[5]\,\Xi x^{(0,0,1)}[x,t,u[0]] - u[2]^2 H^{(0,0,2)}[x,t,u[0]] + u[2]^2 u[5]\,\Xi t^{(0,0,2)}[x,t,u[0]] + u[2]^3\,\Xi x^{(0,0,2)}[x,t,u[0]] + H^{(0,1,0)}[x,t,u[0]] - u[5]\,\Xi t^{(0,1,0)}[x,t,u[0]] - u[2]\,\Xi x^{(0,1,0)}[x,t,u[0]] + 2\,u[4]\,\Xi t^{(1,0,0)}[x,t,u[0]] + 2\,u[5]\,\Xi x^{(1,0,0)}[x,t,u[0]] - 2\,u[2]\,H^{(1,0,1)}[x,t,u[0]] + 2\,u[2]\,u[5]\,\Xi t^{(1,0,1)}[x,t,u[0]] + 2\,u[2]^2\,\Xi x^{(1,0,1)}[x,t,u[0]] - H^{(2,0,0)}[x,t,u[0]] + u[5]\,\Xi t^{(2,0,0)}[x,t,u[0]] + u[2]\,\Xi x^{(2,0,0)}[x,t,u[0]]\}$$

```
(* Coefficients of the polynomial in u[...] should be zero *)
Column[GetConditionsForPointSymmetries[zero, Table[u[j], {j, 1, 5}]]]
```

$$2\,\Xi t^{(0,0,1)}[x,t,u[0]]$$
$$\Xi t^{(0,0,2)}[x,t,u[0]]$$
$$\Xi x^{(0,0,2)}[x,t,u[0]]$$
$$2\,\Xi t^{(1,0,0)}[x,t,u[0]]$$
$$2\left(\Xi x^{(0,0,1)}[x,t,u[0]] + \Xi t^{(1,0,1)}[x,t,u[0]]\right)$$
$$-H^{(0,0,2)}[x,t,u[0]] + 2\,\Xi x^{(1,0,1)}[x,t,u[0]]$$
$$H^{(0,1,0)}[x,t,u[0]] - H^{(2,0,0)}[x,t,u[0]]$$
$$-\Xi t^{(0,1,0)}[x,t,u[0]] + 2\,\Xi x^{(1,0,0)}[x,t,u[0]] + \Xi t^{(2,0,0)}[x,t,u[0]]$$
$$-\Xi x^{(0,1,0)}[x,t,u[0]] - 2\,H^{(1,0,1)}[x,t,u[0]] + \Xi x^{(2,0,0)}[x,t,u[0]]$$

Second ansatz

```
ξ[t] = τ[t];
ξ[x] = 1/2 D[τ[t], t] x + χ[t];
η[u] = α[x, t] u[x, t] + β[x, t];
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

```
u[0] = u[x, t]
u[1] = u^(0,1)[x, t]
u[2] = u^(1,0)[x, t]
u[3] = u^(0,2)[x, t]
u[4] = u^(1,1)[x, t]
u[5] = u^(2,0)[x, t]
```

$$\left\{-u[2]\left(\chi'[t] + \frac{1}{2}\,x\,\tau''[t]\right) + u[0]\,\alpha^{(0,1)}[x,t] + \beta^{(0,1)}[x,t] - 2\,u[2]\,\alpha^{(1,0)}[x,t] - u[0]\,\alpha^{(2,0)}[x,t] - \beta^{(2,0)}[x,t]\right\}$$

```
Column[GetConditionsForPointSymmetries[zero, Table[u[j], {j, 0, 5}]]]
```

$-2\chi'[t] - x\,\tau''[t] - 4\,\alpha^{(1,0)}[x, t]$
$2\left(\alpha^{(0,1)}[x, t] - \alpha^{(2,0)}[x, t]\right)$
$2\left(\beta^{(0,1)}[x, t] - \beta^{(2,0)}[x, t]\right)$

Third ansatz

```
ξ[t] = τ[t];
ξ[x] = 1/2 D[τ[t], t] x + χ[t];
η[u] = (-1/8 D[τ[t], t, t] x^2 - 1/2 D[χ[t], t] x + γ[t]) u[x, t] + β[x, t];
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

u[0] = u[x, t]
u[1] = $u^{(0,1)}[x, t]$
u[2] = $u^{(1,0)}[x, t]$
u[3] = $u^{(0,2)}[x, t]$
u[4] = $u^{(1,1)}[x, t]$
u[5] = $u^{(2,0)}[x, t]$

$\left\{u[0]\,\gamma'[t] + \frac{1}{4}u[0]\,\tau''[t] - \frac{1}{2}x\,u[0]\,\chi''[t] - \frac{1}{8}x^2 u[0]\,\tau^{(3)}[t] + \beta^{(0,1)}[x, t] - \beta^{(2,0)}[x, t]\right\}$

```
Column[GetConditionsForPointSymmetries[zero, Flatten[{Table[u[j], {j, 0, 5}], x}] ]]
```

$2\left(4\gamma'[t] + \tau''[t]\right)$
$-4\chi''[t]$
$-\tau^{(3)}[t]$
$8\left(\beta^{(0,1)}[x, t] - \beta^{(2,0)}[x, t]\right)$

The last ansatz

```
ξ[x] = c[1] + c[4] x + 2 c[5] t + 4 c[6] x t;
ξ[t] = c[2] + 2 c[4] t + 4 c[6] t^2;
η[u] = (c[3] - c[5] x - 2 c[6] t - c[6] x^2) u[x, t] + β[x, t];
zero = CheckPointSymmetryOfDE[PDEs, subs, IndepVar, DepVar, ξ, η]
```

u[0] = u[x, t]
u[1] = $u^{(0,1)}[x, t]$
u[2] = $u^{(1,0)}[x, t]$
u[3] = $u^{(0,2)}[x, t]$
u[4] = $u^{(1,1)}[x, t]$
u[5] = $u^{(2,0)}[x, t]$

$\left\{\beta^{(0,1)}[x, t] - \beta^{(2,0)}[x, t]\right\}$

Infinitesimal generators, point transformations and commutator table from the last ansatz

```
ShowPointSymmetriesAndCommutationRelations[
 X, f, ϵ, IndepVar, DepVar, ξ, η, c, 6, {β[x, t] → 0}]
```

Infinitesimal operators:

$X[1]f[x,t,u] = f^{(1,0,0)}[x,t,u]$

$X[2]f[x,t,u] = f^{(0,1,0)}[x,t,u]$

$X[3]f[x,t,u] = u\,f^{(0,0,1)}[x,t,u]$

$X[4]f[x,t,u] = 2\,t\,f^{(0,1,0)}[x,t,u] + x\,f^{(1,0,0)}[x,t,u]$

$X[5]f[x,t,u] = -u\,x\,f^{(0,0,1)}[x,t,u] + 2\,t\,f^{(1,0,0)}[x,t,u]$

$X[6]f[x,t,u] = u\left(-2\,t - x^2\right) f^{(0,0,1)}[x,t,u] + 4\,t^2\,f^{(0,1,0)}[x,t,u] + 4\,t\,x\,f^{(1,0,0)}[x,t,u]$

... Solve: Inconsistent or redundant transcendental equation. After reduction, the bad equation is $-x^2 + 2\,t\,\text{InverseFunction}[\#1\,e^{\#1}\,\&,1,1]\left[-\frac{e^{-\frac{C[\ll 1\gg]^2}{2\,C[\ll 1\gg]}}\,C[2]^2}{2\,C[1]}\right] == 0.$

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Corresponding global transformations:

$X[1]$ gives $\{x[\epsilon] \to x + \epsilon,\ t[\epsilon] \to t,\ u[\epsilon] \to u\}$

$X[2]$ gives $\{x[\epsilon] \to x,\ t[\epsilon] \to t + \epsilon,\ u[\epsilon] \to u\}$

$X[3]$ gives $\{x[\epsilon] \to x,\ t[\epsilon] \to t,\ u[\epsilon] \to u\,e^{\epsilon}\}$

$X[4]$ gives $\{x[\epsilon] \to x\,e^{\epsilon},\ t[\epsilon] \to t\,e^{2\epsilon},\ u[\epsilon] \to u\}$

$X[5]$ gives $\{t[\epsilon] \to t,\ x[\epsilon] \to x + 2\,t\,\epsilon,\ u[\epsilon] \to u\,e^{-\epsilon\,(x+t\,\epsilon)}\}$

$X[6]$ gives $\left\{t[\epsilon] \to \frac{t}{1-4\,t\,\epsilon},\ x[\epsilon] \to \frac{x}{1-4\,t\,\epsilon},\ u[\epsilon] \to u\,e^{-\frac{x^2}{4\,t-16\,t^2\,\epsilon}}\sqrt{-t\,e^{\frac{x^2}{2\,t}}}\sqrt{-\frac{1}{t}+4\,\epsilon}\right\}$

Commutator table:

	1	2	3	4	5	6
1	0	0	0	X[1]	-X[3]	2 X[5]
2	0	0	0	2 X[2]	2 X[1]	-2 X[3] + 4 X[4]
3	0	0	0	0	0	0
4	-X[1]	-2 X[2]	0	0	X[5]	2 X[6]
5	X[3]	-2 X[1]	0	-X[5]	0	0
6	-2 X[5]	2 X[3] - 4 X[4]	0	-2 X[6]	0	0